

Mathematical Models for Interacting Populations

Prey-predator models

Volterra (1926) first proposed a simple model for the predation of one species by another to explain the oscillatory levels of certain fish catches in the Adriatic.

$x(t)$ - the prey population at the moment t

$y(t)$ - the predator population at the moment t

Assumptions:

- the prey in the absence of any predation grows unboundedly as in the Malthus' model (unlimited resources)
- the effect of the predation is to reduce the prey's per capita growth rate by a term proportional to the number of interactions between the two populations ($x y$)
- in the absence of any prey for sustenance the predator's death rate results in exponential decay
- the prey's contribution to the predators' growth rate is proportional to the available prey as well as to the size of the predator population, this is proportional to the number of interactions between the two populations

$$\frac{d}{dt} x(t) = a x(t) - b x(t) y(t)$$

$$\frac{d}{dt} y(t) = -c y(t) + d x(t) y(t)$$

$$x(0) = x_0$$

$$y(0) = y_0$$

where a, b, c, d are positive.

This is known as the Lotka–Volterra model since the same equations were also derived by Lotka (1920; see also 1925) from a hypothetical chemical reaction which he said could exhibit periodic behaviour in the chemical concentrations.

```
> restart:with(DEtools):
```

```
> with(linalg):
```

```
Warning, the name adjoint has been redefined
```

```
Warning, the protected names norm and trace have been redefined and unprotected
```

```
> with(plots):
```

```
Warning, the name changecoords has been redefined
```

```
> eq1:=diff(x(t),t) =a*x(t)-b*x(t)*y(t);
```

$$eq1 := \frac{d}{dt} x(t) = a x(t) - b x(t) y(t)$$

```

> eq2:=diff(y(t),t) =-c*y(t)+d*x(t)*y(t);
      eq2 :=  $\frac{d}{dt} y(t) = -c y(t) + d x(t) y(t)$ 

> pp_syst:=eq1,eq2;
pp_syst :=  $\frac{d}{dt} x(t) = a x(t) - b x(t) y(t), \frac{d}{dt} y(t) = -c y(t) + d x(t) y(t)$ 

> f1:=(x,y)->a*x-b*x*y;
      f1 := (x, y)  $\rightarrow a x - b x y$ 

> f2:=(x,y)->-c*y+d*x*y;
      f2 := (x, y)  $\rightarrow -c y + d x y$ 

> EqP:=solve({f1(x,y)=0,f2(x,y)=0},{x,y});
      EqP := {x=0, y=0}, {x= $\frac{c}{d}$ , y= $\frac{a}{b}$ }

> EqP[1,1];
      x=0

> J:=jacobian([f1(x,y),f2(x,y)],[x,y]);
      J :=  $\begin{bmatrix} a - b y & -x b \\ y d & -c + d x \end{bmatrix}$ 

> J1:=subs(EqP[1,1],EqP[1,2],eval(J));
      J1 :=  $\begin{bmatrix} a & 0 \\ 0 & -c \end{bmatrix}$ 

> eigenvals(J1);
      a, -c

> J2:=subs(EqP[2,1],EqP[2,2],eval(J));
      J2 :=  $\begin{bmatrix} 0 & -\frac{c b}{d} \\ \frac{a d}{b} & 0 \end{bmatrix}$ 

> eigenvals(J2);
       $\sqrt{-c a}, -\sqrt{-c a}$ 

> deq:=diff(y(x),x)=f2(x,y(x))/f1(x,y(x));
      deq :=  $\frac{d}{dx} y(x) = \frac{-c y(x) + d x y(x)}{a x - b x y(x)}$ 

> dsolve(deq,y(x),implicit);
      d x - c ln(x) + b y(x) - a ln(y(x)) + _C1 = 0

> a:=0.8;b:=0.2;c:=0.4;d:=0.2;x0:=8;y0:=5;
      a := 0.8
      b := 0.2

```

$c := 0.4$

$d := 0.2$

$x0 := 8$

$y0 := 5$

> **EqP;**

$\{x = 0, y = 0\}, \{x = 2.000000000, y = 4.000000000\}$


>

gx:=DEplot([pp_syst],[x(t),y(t)],t=0..50,[[x(0)=x0,y(0)=y0]],stepsize=0.1,linecolor=blue,scene=[t,x(t)]):

>


gy:=DEplot([pp_syst],[x(t),y(t)],t=0..50,[[x(0)=x0,y(0)=y0]],stepsize=0.1,linecolor=red,scene=[t,y(t)]):

> **display(gx,gy);**




>

DEplot([pp_syst],[x(t),y(t)],t=0..50,[[x(0)=x0,y(0)=y0]],stepsize=0.1,linecolor=blue);



```
>
DEplot([pp_syst],[x(t),y(t)],t=0..50,[[x(0)=x0,y(0)=y0],[x(0)=2,y(0)=
2],[x(0)=2,y(0)=3]],stepsize=0.1,linecolor=blue);
```



Since the orbits are closed then the solution are periodically with some period T . The average values of $x(t)$ and $y(t)$ are defined as

$$x_{average} = \frac{\int_0^T x(t) dt}{T} \qquad y_{average} = \frac{\int_0^T y(t) dt}{T}$$

It can be proved that average values of $x(t)$ and $y(t)$ are equal with the positive equilibrium point coordinates

$$x_{average} = \frac{a}{b} \qquad y_{average} = \frac{c}{d}$$

In the mid 1920's the Italian biologist Umberto D'Ancona was studying the population variations of various species of fish that interact ,with each other. In the course of his research, he came across some data on percentages-of-total-catch of several species of fish that were caught into different Mediterranean ports in the years that spanned World War I. In particular, the data gave the percentage-of-total-catch of selachians, (sharks, skates, rays, etc.) which are not very desirable as food fish. D'Ancona was puzzled by the very large increase in the percentage of selachians during the period of the war. Obviously, he reasoned, the increase in the percentage of selachians was due to the greatly reduced level of fishing during this period. But how does the intensity of fishing affect the fish populations?

Let's include the effects of harvesting in the prey-predator model. Observe that fishing decreases the population of food fish at a rate $E x(t)$, and decreases the population of selachians at a rate $E y(t)$. The constant E reflects the intensity of fishing; i.e., the number of boats at sea and the number of nets in the water. Thus, the true state of affairs is described by the system

$$\frac{d}{dt}x(t) = a x(t) - b x(t) y(t) - E x(t)$$

$$\frac{d}{dt}y(t) = -c y(t) + d x(t) y(t) - E y(t)$$

or

$$\frac{d}{dt}x(t) = (a - E) x(t) - b x(t) y(t)$$

$$\frac{d}{dt}y(t) = -(c + E) y(t) + d x(t) y(t)$$

> **a:='a';b:='b';c:='c';d:='d';**

a := a

b := b

c := c

d := d

> **f1:=(x,y)->(a-E)*x-b*x*y;**

f1 := (x, y) → (a - E) x - b x y

> **f2:=(x,y)->-(c+E)*y+d*x*y;**

f2 := (x, y) → -(c + E) y + d x y

> **EqP:=solve({f1(x,y)=0,f2(x,y)=0},{x,y});**

EqP := {x=0, y=0}, {x = $\frac{c+E}{d}$, y = $-\frac{-a+E}{b}$ }

Hence, the average values of x(t) and y(t) are now

$$x_{average} = \frac{c + E}{d} \quad y_{average} = \frac{a - E}{b}$$

Consequently, a moderate amount of fishing ($E < a$) actually increases the number of food fish, on the average, and decreases the number of selachians (predators). Conversely, a reduced level of fishing increases the number of selachians, on the average, and decreases the number of food fish. This remarkable result, which is known as Volterra's principle, explains the data of D' Ancona.

> **eq1:=diff(x(t),t) =(a-E)*x(t)-b*x(t)*y(t);**

eq1 := $\frac{d}{dt}x(t) = (a - E) x(t) - b x(t) y(t)$

> **eq2:=diff(y(t),t) =-(c+E)*y(t)+d*x(t)*y(t);**

$$eq2 := \frac{d}{dt} y(t) = -(c + E) y(t) + d x(t) y(t)$$

```
> a:=0.8;b:=0.2;c:=0.4;d:=0.2;x0:=8;y0:=5;E:=0.3;
```

```
    a := 0.8
```

```
    b := 0.2
```

```
    c := 0.4
```

```
    d := 0.2
```

```
    x0 := 8
```

```
    y0 := 5
```

```
    E := 0.3
```

```
> h_syst:=eq1,eq2;
```

```
h_syst :=  $\frac{d}{dt} x(t) = 0.5 x(t) - 0.2 x(t) y(t)$ ,  $\frac{d}{dt} y(t) = -0.7 y(t) + 0.2 x(t) y(t)$ 
```


```
>
```

```
gx:=DEplot([h_syst],[x(t),y(t)],t=0..50,[[x(0)=x0,y(0)=y0]],stepsize=0.1,linecolor=blue,scene=[t,x(t)]):
```

```
>
```


```
gy:=DEplot([h_syst],[x(t),y(t)],t=0..50,[[x(0)=x0,y(0)=y0]],stepsize=0.1,linecolor=red,scene=[t,y(t)]):
```

```
> display(gx,gy);
```



```
>
```

```
DEplot([h_syst],[x(t),y(t)],t=0..50,[[x(0)=x0,y(0)=y0],[x(0)=2,y(0)=2],[x(0)=2,y(0)=4]],stepsize=0.1,linecolor=blue);
```



> **E:=0.6;**

$E := 0.6$

> **h_syst:=eq1,eq2;**

$$h_syst := \frac{d}{dt} x(t) = 0.2 x(t) - 0.2 x(t) y(t), \frac{d}{dt} y(t) = -1.0 y(t) + 0.2 x(t) y(t)$$


>

gx:=DEplot([h_syst],[x(t),y(t)],t=0..50,[[x(0)=x0,y(0)=y0]],stepsize=0.1,linecolor=blue,scene=[t,x(t)]):

>


gy:=DEplot([h_syst],[x(t),y(t)],t=0..50,[[x(0)=x0,y(0)=y0]],stepsize=0.1,linecolor=red,scene=[t,y(t)]):

> **display(gx,gy);**



>

DEplot([h_syst],[x(t),y(t)],t=0..50,[[x(0)=x0,y(0)=y0],[x(0)=2,y(0)=2],[x(0)=2,y(0)=4]],stepsize=0.1,linecolor=blue);



More realistic prey-predator models

Prey-predator model with intraspecies competition

Assumptions:

- the prey in the absence of any predation grows boundedly as in the logistical model (limited resources)
- the effect of the predation is to reduce the prey's per capita growth rate by a term proportional to the number of interactions between the two populations ($x y$)
- in the absence of any prey for sustenance the predator's death rate results in exponential decay
- the prey's contribution to the predators' growth rate is proportional to the available prey as well as to the size of the predator population, this is proportional to the number of interactions between the two populations
- in the presence of the prey there exists a competitions between the predators for the prey, this is proportional with the number of interaction of predators y^2

$$\frac{d}{dt} x(t) = r x(t) \left(1 - \frac{x(t)}{K} \right) - \alpha x(t) y(t)$$

$$\frac{d}{dt} y(t) = -\gamma y(t) + \delta x(t) y(t) - \beta y(t)^2$$

$$x(0) = x_0$$

$$y(0) = y_0$$

where $r, K, \alpha, \gamma, \delta, \beta$ are positive.

```
> restart:with(DEtools):
```

```
> with(linalg):
```

```
Warning, the name adjoint has been redefined
```

```
Warning, the protected names norm and trace have been redefined and unprotected
```

```
> with(plots):
```


Warning, the name changecoords has been redefined

> **eq1:=diff(x(t),t)=r*x(t)*(1-x(t)/K)-alpha*x(t)*y(t);**

$$eq1 := \frac{d}{dt} x(t) = r x(t) \left(1 - \frac{x(t)}{K}\right) - \alpha x(t) y(t)$$

> **eq2:=diff(y(t),t)=-gamma*y(t)+delta*x(t)*y(t)-beta*y(t)^2;**

$$eq2 := \frac{d}{dt} y(t) = -\gamma y(t) + \delta x(t) y(t) - \beta y(t)^2$$

> **pp_syst:=eq1,eq2;**

pp_syst :=

$$\frac{d}{dt} x(t) = r x(t) \left(1 - \frac{x(t)}{K}\right) - \alpha x(t) y(t), \frac{d}{dt} y(t) = -\gamma y(t) + \delta x(t) y(t) - \beta y(t)^2$$

> **f1:=(x,y)->r*x*(1-x/K)-alpha*x*y;**

$$f1 := (x, y) \rightarrow r x \left(1 - \frac{x}{K}\right) - \alpha x y$$

> **f2:=(x,y)->-gamma*y+delta*x*y-beta*y^2;**

$$f2 := (x, y) \rightarrow -\gamma y + \delta x y - \beta y^2$$

> **EqP:=solve({f1(x,y)=0,f2(x,y)=0},{x,y});**

$$EqP := \{x=0, y=0\}, \{x=K, y=0\}, \{x=0, y=-\frac{\gamma}{\beta}\},$$

$$\left\{y = \frac{r(-\gamma + \delta K)}{\delta K \alpha + \beta r}, x = \frac{K(\beta r + \alpha \gamma)}{\delta K \alpha + \beta r}\right\}$$

The equilibrium point $(0; -\frac{\gamma}{\beta})$ is not positive, so it is not realistic and the equilibrium point

$(\frac{K(\beta r + \alpha \gamma)}{\delta K \alpha + \beta r}, \frac{r(-\gamma + \delta K)}{\delta K \alpha + \beta r})$ is positive iff $\delta K - \gamma > 0$.

$$y = 0$$

> **J:=jacobian([f1(x,y),f2(x,y)], [x,y]);**

$$J := \begin{bmatrix} r \left(1 - \frac{x}{K}\right) - \frac{rx}{K} - \alpha y & -\alpha x \\ y \delta & -\gamma + \delta x - 2\beta y \end{bmatrix}$$

> **J1:=subs(EqP[1,1],EqP[1,2],eval(J));**

$$J1 := \begin{bmatrix} r & 0 \\ 0 & -\gamma \end{bmatrix}$$

> **eigenvals(J1);**

$$r, -\gamma$$

Since $r > 0$ this implies that $(0;0)$ is unstable.

> **J2:=subs(EqP[3,1],EqP[3,2],eval(J));**

$$J2 := \begin{bmatrix} r + \frac{\alpha \gamma}{\beta} & 0 \\ -\frac{\gamma \delta}{\beta} & \gamma \end{bmatrix}$$

> **eigenvals(J2);**

$$\frac{\beta r + \alpha \gamma}{\beta}, \gamma$$

If $\delta K - \gamma < 0$ then (K;0) is local asymptotically stable.

If $\delta K - \gamma > 0$ then (K;0) is unstable.

> **J3:=subs(EqP[4,1],EqP[4,2],eval(J));**

$$J3 := \begin{bmatrix} r \left(1 - \frac{\beta r + \alpha \gamma}{\delta K \alpha + \beta r} \right) - \frac{r(\beta r + \alpha \gamma)}{\delta K \alpha + \beta r} - \frac{\alpha r(-\gamma + \delta K)}{\delta K \alpha + \beta r}, & -\frac{\alpha K(\beta r + \alpha \gamma)}{\delta K \alpha + \beta r} \\ \frac{r(-\gamma + \delta K) \delta}{\delta K \alpha + \beta r}, & -\gamma + \frac{\delta K(\beta r + \alpha \gamma)}{\delta K \alpha + \beta r} - \frac{2 \beta r(-\gamma + \delta K)}{\delta K \alpha + \beta r} \end{bmatrix}$$

> **J3[1,1]:=factor(expand(J3[1,1]));**

$$J3_{1,1} := -\frac{r(\beta r + \alpha \gamma)}{\delta K \alpha + \beta r}$$

> **J3[2,2]:=factor(expand(J3[2,2]));**

$$J3_{2,2} := -\frac{\beta r(-\gamma + \delta K)}{\delta K \alpha + \beta r}$$

> **eval(J3);**

$$\begin{bmatrix} -\frac{r(\beta r + \alpha \gamma)}{\delta K \alpha + \beta r} & -\frac{\alpha K(\beta r + \alpha \gamma)}{\delta K \alpha + \beta r} \\ \frac{r(-\gamma + \delta K) \delta}{\delta K \alpha + \beta r} & -\frac{\beta r(-\gamma + \delta K)}{\delta K \alpha + \beta r} \end{bmatrix}$$

> **eigenvals(J3);**

$$\begin{aligned} & (-\delta K \beta r - \alpha r \gamma + \beta r \gamma - \beta r^2 + (\delta^2 K^2 \beta^2 r^2 + 2 r^2 \alpha \gamma \delta K \beta - 2 \delta K \beta^2 r^2 \gamma \\ & - 2 \beta^2 r^3 \delta K + r^2 \alpha^2 \gamma^2 + 2 r^2 \alpha \gamma^2 \beta + 2 r^3 \alpha \gamma \beta + \beta^2 r^2 \gamma^2 + 2 \beta^2 r^3 \gamma + \beta^2 r^4 \\ & - 4 \alpha^2 r \delta^2 K^2 \gamma + 4 \alpha^2 r \delta K \gamma^2 - 4 \alpha r^2 \delta^2 K^2 \beta)^{(1/2)}) / (2 (\delta K \alpha + \beta r)), (-\delta K \beta r \\ & - \alpha r \gamma + \beta r \gamma - \beta r^2 - (\delta^2 K^2 \beta^2 r^2 + 2 r^2 \alpha \gamma \delta K \beta - 2 \delta K \beta^2 r^2 \gamma - 2 \beta^2 r^3 \delta K \\ & + r^2 \alpha^2 \gamma^2 + 2 r^2 \alpha \gamma^2 \beta + 2 r^3 \alpha \gamma \beta + \beta^2 r^2 \gamma^2 + 2 \beta^2 r^3 \gamma + \beta^2 r^4 - 4 \alpha^2 r \delta^2 K^2 \gamma \\ & + 4 \alpha^2 r \delta K \gamma^2 - 4 \alpha r^2 \delta^2 K^2 \beta)^{(1/2)}) / (2 (\delta K \alpha + \beta r)) \end{aligned}$$

> **(charpoly(J3,lambda))=0;**

$$\begin{aligned} & (\alpha \delta K \lambda^2 + \alpha \delta K \gamma r + \alpha r \gamma \lambda - \alpha r \gamma^2 + \delta K \lambda \beta r + \delta K \beta r^2 + \beta r^2 \lambda + \lambda^2 \beta r - \lambda \beta r \gamma \\ & - \beta r^2 \gamma) / (\delta K \alpha + \beta r) = 0 \end{aligned}$$

> **eq1:=(delta*K*alpha+beta*r)*(charpoly(J3,lambda))=0;**

$$eq1 := \alpha \delta K \lambda^2 + \alpha \delta K \gamma r + \alpha r \gamma \lambda - \alpha r \gamma^2 + \delta K \lambda \beta r + \delta K \beta r^2 + \beta r^2 \lambda + \lambda^2 \beta r - \lambda \beta r \gamma - \beta r^2 \gamma = 0$$

> **collect(eq1, lambda);**

$$(\delta K \alpha + \beta r) \lambda^2 + (\delta K \beta r + \alpha r \gamma - \beta r \gamma + \beta r^2) \lambda + \alpha \delta K \gamma r - \alpha r \gamma^2 + \delta K \beta r^2 - \beta r^2 \gamma = 0$$

> **a1:=coeff(lhs(eq1), lambda, 2);**

$$a1 := \delta K \alpha + \beta r$$

> **a2:=factor(coeff(lhs(eq1), lambda, 1));**

$$a2 := r(\delta K \beta + \alpha \gamma - \beta \gamma + \beta r)$$

> **a3:=factor(coeff(lhs(eq1), lambda, 0));**

$$a3 := (-\gamma + \delta K) r(\beta r + \alpha \gamma)$$

The necessary and sufficient condition to have $\text{Re}(\lambda) < 0$ is $\frac{a2}{a1} < 0$ and $\frac{a3}{a1} > 0$. In our case $a1 > 0$, thus if $a2 < 0$ and $a3 > 0$ then $\text{Re}(\lambda) < 0$.

> **a2<0;**

$$-r(-\gamma \alpha - \delta K \beta - r \beta + \beta \gamma) < 0$$

> **a3>0;**

$$0 < -r(\gamma - \delta K)(r \beta + \gamma \alpha)$$

$a2 < 0$ implies that $\frac{r}{\gamma} + \frac{\delta K}{\gamma} + \frac{\alpha}{\beta} < 1$

$a3 > 0$ implies that $\delta K - \gamma > 0$

Let's consider the case $r = 0.3$, $K = 5$, $\alpha = 0.2$, $\gamma = 0.6$, $\delta = 0.1$, $\beta = 0.2$, $x_0 = 3$, $y_0 = 1$

> **f1:=(x,y,r,K,alpha)->r*x*(1-x/K)-alpha*x*y;**

$$f1 := (x, y, r, K, \alpha) \rightarrow r x \left(1 - \frac{x}{K}\right) - \alpha x y$$

> **f2:=(x,y,gamma,delta,beta)->-gamma*y+delta*x*y-beta*y^2;;**

$$f2 := (x, y, \gamma, \delta, \beta) \rightarrow -\gamma y + \delta x y - \beta y^2$$

> **eq1:=diff(x(t),t)=f1(x(t),y(t),0.3,5,0.2);;**

$$eq1 := \frac{d}{dt} x(t) = 0.3 x(t) \left(1 - \frac{1}{5} x(t)\right) - 0.2 x(t) y(t)$$

> **eq2:=diff(y(t),t)=f2(x(t),y(t),0.6,0.1,0.2);**

$$eq2 := \frac{d}{dt} y(t) = -0.6 y(t) + 0.1 x(t) y(t) - 0.2 y(t)^2$$


> **syst1:=eq1,eq2;**

$$\text{syst1} := \frac{d}{dt} x(t) = 0.3 x(t) \left(1 - \frac{1}{5} x(t) \right) - 0.2 x(t) y(t),$$

$$\frac{d}{dt} y(t) = -0.6 y(t) + 0.1 x(t) y(t) - 0.2 y(t)^2$$

```
> EqP:=solve({f1(x,y,0.3,5,0.2)=0,f2(x,y,0.6,0.1,0.2)=0},{x,y});
EqP := {x=0., y=0.}, {y=-3., x=0.}, {x=5., y=0.},
      {y=-0.1875000000, x=5.625000000 }
```


```
>
DEplot([syst1],[x(t),y(t)],t=0..50,[[x(0)=3,y(0)=1]],stepsize=0.1,lin
ecolor=blue);
```



```
>
gx:=DEplot([syst1],[x(t),y(t)],t=0..50,[[x(0)=3,y(0)=1]],stepsize=0.1
,linecolor=blue,scene=[t,x(t)]):
```

```
>
gy:=DEplot([syst1],[x(t),y(t)],t=0..50,[[x(0)=3,y(0)=1]],stepsize=0.1
,linecolor=red,scene=[t,y(t)]):
```

```
> display(gx,gy);
```



Let's consider the case $r = 0.3$, $K = 5$, $\alpha = 0.2$, $\gamma = 0.3$, $\delta = 0.2$, $\beta = 0.2$, $x_0 = 3$, $y_0 = 1$

```
> eq1:=diff(x(t),t)=f1(x(t),y(t),0.3,5,0.2);
```

$$eq1 := \frac{d}{dt} x(t) = 0.3 x(t) \left(1 - \frac{1}{5} x(t) \right) - 0.2 x(t) y(t)$$

```
> eq2:=diff(y(t),t)=f2(x(t),y(t),0.3,0.2,0.2);
```

$$eq2 := \frac{d}{dt} y(t) = -0.3 y(t) + 0.2 x(t) y(t) - 0.2 y(t)^2$$

```
> syst2:=eq1,eq2;
```

$$syst2 := \frac{d}{dt} x(t) = 0.3 x(t) \left(1 - \frac{1}{5} x(t) \right) - 0.2 x(t) y(t),$$

$$\frac{d}{dt} y(t) = -0.3 y(t) + 0.2 x(t) y(t) - 0.2 y(t)^2$$


```
> EqP:=solve({f1(x,y,0.3,5,0.2)=0,f2(x,y,0.3,0.2,0.2)=0},{x,y});
```

```
EqP := {x = 0., y = 0.}, {y = -1.500000000, x = 0.}, {x = 5., y = 0.},
```

```
{y = 0.8076923077, x = 2.307692308 }
```

```
>
```

```
DEplot([syst2],[x(t),y(t)],t=0..50,[[x(0)=3,y(0)=1]],stepsize=0.1,lin  
ecolor=blue);
```




```
>
```

```
gx:=DEplot([syst2],[x(t),y(t)],t=0..50,[[x(0)=3,y(0)=1]],stepsize=0.1  
,linecolor=blue,scene=[t,x(t)]):
```

```
>
```

```
gy:=DEplot([syst2],[x(t),y(t)],t=0..50,[[x(0)=3,y(0)=1]],stepsize=0.1  
,linecolor=red,scene=[t,y(t)]):
```

```
> display(gx,gy);
```



Prey-predator model with food saturation

In case of prey population we can assume that develops according to the logistical model in the absence of the predator and its increasing is reduced by a predation term

$$\frac{d}{dt} x(t) = r x(t) \left(1 - \frac{x(t)}{K} \right) - x(t) y(t) R(x(t))$$

$x R(x)$ is the predation term. The predation term, which is the functional response of the predator to change in the prey density, generally shows some saturation effect instead of a predator response of $b x$ or αx (which are linear) as in the above models. A suitable choice for $R(x)$ is

$$R(x) = \frac{a x}{x + b}$$

In the case of predator equation we can take

$$\frac{d}{dt} y(t) = p y(t) \left(1 - \frac{h y(t)}{x(t)} \right)$$

The term $h \frac{y^2}{x}$ measures the predator intraspecies competition.

Thus we obtain the following prey-predator model (Holling–Tanner model)

$$\frac{d}{dt} x(t) = r x(t) \left(1 - \frac{x(t)}{K} \right) - \frac{a x(t) y(t)}{b + x(t)}$$

$$\frac{d}{dt} y(t) = p y(t) \left(1 - \frac{h y(t)}{x(t)} \right)$$

> **restart:with(DEtools):with(linalg):with(plots):**

Warning, the name adjoint has been redefined

Warning, the protected names norm and trace have been redefined and unprotected

Warning, the name changecoords has been redefined

> **eq1:=diff(x(t),t) = r*x(t)*(1-x(t)/K)-a*x(t)*y(t)/(b+x(t));**

$$eq1 := \frac{d}{dt} x(t) = r x(t) \left(1 - \frac{x(t)}{K} \right) - \frac{a x(t) y(t)}{b + x(t)}$$

> **eq2:=diff(y(t),t) = p*y(t)*(1-h*y(t)/x(t));**

$$eq2 := \frac{d}{dt} y(t) = p y(t) \left(1 - \frac{h y(t)}{x(t)} \right)$$

> **f1:=(x,y)->r*x*(1-x/K)-a*x*y/(b+x);**

$$f1 := (x, y) \rightarrow r x \left(1 - \frac{x}{K} \right) - \frac{a x y}{b + x}$$

> **f2:=(x,y)->p*y*(1-h*y/x);**

$$f2 := (x, y) \rightarrow p y \left(1 - \frac{h y}{x} \right)$$

> **EqP:=solve({f1(x,y)=0,f2(x,y)=0},{x,y});**

$$EqP := \{y = 0, x = K\}, \{y = \text{RootOf}(r h^2 _Z^2 + (-r K h + r h b + a K) _Z - r K b), x = h \text{RootOf}(r h^2 _Z^2 + (-r K h + r h b + a K) _Z - r K b)\}$$

> **eq3:=f1(h*y,y)=0;**

$$eq3 := r h y \left(1 - \frac{h y}{K} \right) - \frac{a h y^2}{b + h y} = 0$$

> **simplify(eq3);**

$$-\frac{h y (-r K b - r K h y + r h y b + r h^2 y^2 + a y K)}{K (b + h y)} = 0$$

> **solve(eq3,y);**

$$0, \frac{r K h - r h b - a K + \sqrt{r^2 K^2 h^2 + 2 r^2 h^2 K b - 2 r K^2 h a + r^2 h^2 b^2 + 2 r h b a K + a^2 K^2}}{2 r h^2}, \frac{r K h - r h b - a K - \sqrt{r^2 K^2 h^2 + 2 r^2 h^2 K b - 2 r K^2 h a + r^2 h^2 b^2 + 2 r h b a K + a^2 K^2}}{2 r h^2}$$

> **a:=1;h:=4;r:=0.8;**

$$a := 1$$

$$h := 4$$

$$r := 0.8$$

> **a/(h*r);**

0.3125000000

> **K:=5;b:=1;p:=0.2;**

$K := 5$

$b := 1$

$p := 0.2$

> **syst:=eq1,eq2;**

$$\text{syst} := \frac{d}{dt} x(t) = 0.8 x(t) \left(1 - \frac{1}{5} x(t)\right) - \frac{x(t) y(t)}{1 + x(t)}, \frac{d}{dt} y(t) = 0.2 y(t) \left(1 - \frac{4 y(t)}{x(t)}\right)$$

> **f1(x,y);f2(x,y);**

$$0.8 x \left(1 - \frac{x}{5}\right) - \frac{x y}{1 + x}$$

$$0.2 y \left(1 - \frac{4 y}{x}\right)$$

> **EqP:=solve({f1(x,y)=0,f2(x,y)=0},{x,y});**

$\text{EqP} := \{y = 0., x = 5.\}, \{y = -0.3319713354, x = -1.327885341\},$
 $\{x = 3.765385341, y = 0.9413463354\}$

> **J:=jacobian([f1(x,y),f2(x,y)],[x,y]);**

$$J := \begin{bmatrix} 0.8 - 0.3200000000 x - \frac{y}{1+x} + \frac{x y}{(1+x)^2} & -\frac{x}{1+x} \\ \frac{0.8 y^2}{x^2} & 0.2 - \frac{1.6 y}{x} \end{bmatrix}$$

> **J1:=subs(EqP[1,1],EqP[1,2],eval(J));**

$$J1 := \begin{bmatrix} -0.8000000000 & -0.8333333333 \\ 0. & 0.2 \end{bmatrix}$$

> **eigenvals(J1);**

-0.8000000000, 0.2000000000

> **J2:=subs(EqP[3,1],EqP[3,2],eval(J));**


$$J2 := \begin{bmatrix} -0.4463760629 & -0.7901533814 \\ 0.05000000002 & -0.2000000000 \end{bmatrix}$$

> **eigenvals(J2);**

-0.3231880314 + 0.1559883906 I, -0.3231880314 - 0.1559883906 I

>


DEplot([syst],[x(t),y(t)],t=0..50,[[x(0)=2,y(0)=4]],stepsize=0.1,line color=blue);



```

>
g1:=DEplot([syst],[x(t),y(t)],t=0..50,[[x(0)=2,y(0)=4]],stepsize=0.1,
linecolor=blue,scene=[t,x(t)]):
>
g2:=DEplot([syst],[x(t),y(t)],t=0..50,[[x(0)=2,y(0)=4]],stepsize=0.1,
linecolor=red,scene=[t,y(t)]):
> display(g1,g2);

```



```

> a:=1;h:=0.5;r:=1;

```

```

a := 1
h := 0.5
r := 1

```

```

> a/(h*r);

```

```

2.000000000

```

```

> K:=5;b:=1;p:=0.2;

```

```

K := 5
b := 1
p := 0.2

```

```

>
> syst:=eq1,eq2;

```

$$\text{syst} := \frac{d}{dt} x(t) = x(t) \left(1 - \frac{1}{5} x(t) \right) - \frac{x(t) y(t)}{1 + x(t)}, \frac{d}{dt} y(t) = 0.2 y(t) \left(1 - \frac{0.5 y(t)}{x(t)} \right)$$

> **f1(x,y);f2(x,y);**

$$x \left(1 - \frac{x}{5} \right) - \frac{xy}{1+x}$$

$$0.2 y \left(1 - \frac{0.5 y}{x} \right)$$

> **EqP:=solve({f1(x,y)=0,f2(x,y)=0},{x,y});**

EqP := {y = 0., x = 5.}, {x = -6.741657387, y = -13.48331477},
 {y = 1.483314774, x = 0.7416573868 }

> **J:=jacobian([f1(x,y),f2(x,y)],[x,y]);**

$$J := \begin{bmatrix} 1 - \frac{2x}{5} - \frac{y}{1+x} + \frac{xy}{(1+x)^2} & -\frac{x}{1+x} \\ \frac{0.10 y^2}{x^2} & 0.2 - \frac{0.20 y}{x} \end{bmatrix}$$

> **J1:=subs(EqP[1,1],EqP[1,2],eval(J));**

$$J1 := \begin{bmatrix} -1.000000000 & -0.8333333333 \\ 0. & 0.2 \end{bmatrix}$$

> **eigenvals(J1);**

-1., 0.2000000000

> **J2:=subs(EqP[3,1],EqP[3,2],eval(J));**


$$J2 := \begin{bmatrix} 0.2143381587 & -0.4258342613 \\ 0.4000000003 & -0.2000000000 \end{bmatrix}$$

> **eigenvals(J2);**

0.007169079350 + 0.3569519256 I, 0.007169079350 - 0.3569519256 I

>

DEplot([syst],[x(t),y(t)],t=0..1000,[[x(0)=5,y(0)=3]],stepsize=0.1,linicolor=blue);




>

```
g1:=DEplot([syst],[x(t),y(t)],t=0..200,[[x(0)=5,y(0)=3]],stepsize=0.1,
,linecolor=blue,scene=[t,x(t)]):
```

>


```
g2:=DEplot([syst],[x(t),y(t)],t=0..200,[[x(0)=5,y(0)=3]],stepsize=0.1,
,linecolor=red,scene=[t,y(t)]):
```

```
> display(g1,g2);
```



>

```
DEplot([syst],[x(t),y(t)],t=0..1000,[[x(0)=0.7,y(0)=1.4]],stepsize=0.1,
,linecolor=blue);
```



>

```
g1:=DEplot([syst],[x(t),y(t)],t=0..300,[[x(0)=0.7,y(0)=1.4]],stepsize=0.1,
,linecolor=blue,scene=[t,x(t)]):
```

>

```
g2:=DEplot([syst],[x(t),y(t)],t=0..300,[[x(0)=0.7,y(0)=1.4]],stepsize=0.1,
,linecolor=red,scene=[t,y(t)]):
```

```
> display(g1,g2);
```

